

Microsoft Excel - DSUM Function



DSUM Function

Adds numbers based on criteria stored in a separate range. The criteria can be as complex as required.

DSUM(database,field,criteria)

Where *database* is a range of cells in data format, *field* indicates which column in *database* is to be used in the calculation and *criteria* is a reference to a range of cells where the criteria values are entered.

The *field* argument can be specified either by using the column label as a literal text value, such as "Units" or "Country", enclosed in double quotes and with the exact same spelling as the column label in the first row of *database* (watch out for rogue space characters) or they can be specified as column index values, i.e. the first column in the list is 1 etc.

The *criteria* argument must be a reference to a separate range of cells, you can not contain the criteria values inside the formula. The range must consist of at least two cells and be arranged in data format. The criteria range can be expanded to encompass any set of criteria values and it can be external to the worksheet where the formula is entered.

Column label
Criterion

Column label	Column label
Criterion	Criterion

Additional criteria are arranged like this, going across the worksheet and keeping the values in adjacent cells.

A single field criterion is arranged as above with the column label over the criterion value.

	A	B	C	D	E
1	Country	Product	Units	Value£	M
2	UK	Soap	1,789	81,460	Ja
3	France	Toothpaste	1,217	11,599	F€
4	Netherlands	Shave	637	63,337	M
5	Denmark	Laundry	713	97,982	A€
6	Greece	Kitchen	991	23,378	M
7	Norway	Soap	1,606	56,327	Ju
8	Germany	Toothpaste	46	11,101	Ju
9	Spain	Shave	888	87,187	A€
10					
11	Country	Units			
12	UK	>500			
13					
14	=DSUM(A1:E9,3,A11:B12)				
15	DSUM(database, field, criteria)				
16					

In the example, the formula in cell A14 is adding up the Units values in column C where the Country is "UK" in column A and the Units are greater than 500.

The formula identifies column C either using an index value, it is the third column in the range:

=DSUM(A1:E9,3,A11:B12)

Or it is identified by using the column label. It is not identified by the letter C:

=DSUM(A1:E9,"Units",A11:B12)

It is bad practice to place the *criteria* range under the *database* range as

in the illustration, there is no room to add extra data to the list.

For more information on how to enter criteria expressions see Excel Criteria Expressions in the next section.

When you add additional criteria values to the range on the worksheet you must also remember to expand the definition of the *criteria* range in the formula, similarly you must reduce the range if you remove criteria. To contain the criteria within just two cells and still produce complex criteria, use a *calculated criteria*.

Calculated criteria in database formulas

Calculated criteria using a two cell criteria range is a much neater implementation than the ever-changing multiple cell criteria range for database formulas. But you must comply with some strict rules to get them to work properly:

- Do not use an existing column label in the top cell of the criteria range; either leave it empty or type-in the word "Calc".
- When you refer to a column in the *database*; refer to the first data cell in that column, *the key cell* and use a relative reference.
- Any other cell references in the formula must be absolute.
- The formula must evaluate to TRUE or FALSE

In the example, cell reference A12 is a calculated criteria and contains the following formula:

=AND(A2="UK",B2="Soap",D2>100)

The DSUM formula is adding the Units where the Country is "UK", the Product is "Soap" and the Value is greater than 100. The entire criteria is contained in one cell.

Database formulas certainly would be much easier to work with if the criteria could held inside the formula itself. But that is not the case; the criteria values must always be entered into separate cells and the reference to those cells entered into the formula. This gets on everyone's nerves!

	A	B	C	D
1	Country	Product	Units	Value
2	UK	Soap	1,789	81,4
3	France	Toothpaste	1,217	11,5
4	Netherlands	Shave	637	63,3
5	UK	Laundry	713	97,9
6	Greece	Kitchen	991	23,3
7	Norway	Soap	1,606	56,3
8	Germany	Toothpaste	46	11,1
9	Spain	Shave	888	87,1
10				
11	Calc			
12	TRUE			
13				
14	=DSUM(A1:E9,"Units",A11:A12)			
15	DSUM(database, field, criteria)			

You can perform similar types of aggregation calculations using Excel Pivot tables, they are much easier to do but they are not formulas and do not recalculate automatically as database formulas do, see Pivot Tables Level 3 manual.

Other database functions such as DCOUNT, DCOUNTA, DMAX, DMIN, DPRODUCT, DAVERAGE etc. take the same form.

When you have many database formulas to enter it can take some time to set up all the corresponding criteria ranges for the formulas to refer to. For a macro to automate this process see Setting up criteria for Database formulas.

Excel Criteria Expressions

Excel criteria expressions are entered as arrangements of values in separate worksheet cells which are used to specify the criteria values for Database functions and Advanced Filters. You refer to the cells containing the criteria expression in the formula or filter.

Multiple conditions in a field

An OR condition. For two or more conditions for a single field, enter the criteria directly below each other. The example specifies those rows that contain either "France", "UK" or "Germany" in the Country column.

Country
France
UK
Germany

One condition in two or more fields

Country	Product	Units
UK	Soap	>250

An AND condition. To specify data that meets one condition in two or more fields, enter all the

criteria in the same row of the criteria range. The example specifies all rows that contain "UK" in the Country column, "Soap" in the Product column and values greater than 250 in the Units column.

One condition in one field or another

Country	Product	Units
UK		
	Soap	
		>250

An OR condition. To specify data that meets either a condition in one column or a condition in another column, enter the criteria in different rows of the criteria

range. The example specifies all rows that contain either "UK" in the Country column, "Soap" in the Product column or values greater than 250 in the Units column.

One of two sets of conditions for two fields

Country	Units
UK	>250
France	>500

An AND condition. To specify data that meets one of two sets of conditions, where each set includes conditions for more than one field, enter the criteria in separate rows. The example specifies all

rows that contain both "UK" in the Country column and values greater than 250 in the Units column and also all rows that contain "France" in the Country column and values greater than 500 in the Units column.

More than two sets of conditions for one field

Units	Units
>2,500	<5,000
<500	

An AND and an OR condition. To specify data that meets greater than two sets of conditions, include multiple columns in the criteria range and repeat the same column heading. The example specifies values that are between 2,500 and 5,000 in addition to values that are less than 500 in the Units column.

Calculated criteria

You can use a calculated value that is the result of a formula. When you use a formula to create a criterion, do not use a column label for the criteria label; either keep the criteria label blank or use a label that is not a column label in the range. It is a convention to just type the word "Calc" into the label cell.

The formula you use for a condition must use a relative reference to refer to the cell that is the first data cell under the header row. All other references in the formula must be absolute and the formula must evaluate to TRUE or FALSE.

	A	B	C	D	E	
1	Country	Product	Units	Value	Month	Calc
2	UK	Soap	1,789	81,460	Jan	=A2="UK"
3	France	Toothpaste	1,217	11,599	Feb	
4	Netherlands	Shave	637	63,337	Mar	
5	Denmark	Laundry	713	97,982	Apr	
6	Germany	Kitchen	991	23,378	May	

The advantage of the calculated criteria is that the criteria definition range can be reduced to two cells and yet quite sophisticated criteria can be expressed.

Examples:

Where the Product is "Soap": **=B2="Soap"**

Where the Product is not "Soap":

=NOT(B2="Soap") or =B2<>"Soap"

Where the country is "UK", "France" or "Germany":

=OR(A2="UK",A2="France",A2="Germany")

Where the Product is "Soap" or "Shave" and the Value is greater than or equal to 2000:

=AND(OR(B2="Soap",B2="Shave"),D2>=2000)

Where the Units figure is above the average of the Units:

=C2>AVERAGE(\$C\$2:\$C\$6)

Where the Value figure is less than the median value:

=D2<MEDIAN(\$D\$2:\$D\$6)

Excel Wildcard Characters

The following wildcard characters can be used as comparison criteria for filters, and when searching and replacing content:

Character	Definition	Example	Finds
? (question mark)	Any single character.	Sm?th	"Smith", "Smyth"
* (asterisk)	Any number of characters.	*can*	Any text containing "can"
~ (tilde)	Followed by ?, *, or ~ treats the characters as literal characters.	Wot~?	Wot?

